Import JJIO
Class Employee
-- Name: A.N. Ominous
-- Does provide for employee objects
Constant Over_Hours = 40.0

Box name ofClass Str is private
Box hours ofType real is private
Box rate ofType real is private

Constructor Employee (n,h,r) is public
Slot n ofClass Str -- name
Slot h ofType real -- hours
Slot r ofType real -- rate
-- Does Create Employee
Set name = n
Set hours = h
Set rate = r
EndConstructor Employee

Function grossPay (none) ofType real is public
Box result ofType real
Box pay ofType real
-- Does return the gross pay
If (hours < Over_Hours) then
Set pay = rate * hours
Else
Set pay = rate * Over_Hours +
rate * 1.5 * (hours - Over_Hours)
EndIf
Set result = pay
EndFunction grossPay

Routine showName (none) is public
-- Does show, display, print the name
Output name
EndRoutine showName

Function worksOverTime (none) ofType bool is public
Box result ofType bool
-- Does employee work overtime
Set result = (hours > Over_Hours)
EndFunction worksOverTime

Routine setRate (amount) is public
Slot amount ofType real
-- Sets pay rate to given amount
Set rate = amount
EndRoutine setRate

Routine PayTest (none) is private
Box first ofClass Employee
Box second ofClass Employee
-- Does test Employee class
Start
New first ofClass Employee with ("Bill Gates", 50.0, 10.00)
New second ofClass Employee with ("John Motil", 10.0, 50.00)
Call second.setRate with (55.0)
Call first.showName
Output " is paid $"
Outputln first.grossPay()
Call second.showName
Output " is paid $"
Outputln second.grossPay()
EndRoutine PayTest

EndClass Employee

-- OUTPUT:
-- Bill Gates is paid $550
-- John Motil is paid $550